

LIGHT PEN



INSTRUCTIONS

CONNECTION - The Colorware Light Pen plugs directly into the 'Right Joystick' input located on the rear of your Radio Shack Color Computer or TDP-100.

TV OR MONITOR RECOMMENDATIONS - The light pen will operate on any Color or B&W screen of 5 inches or larger. A 9" or larger screen will give better results.

Some televisions, especially smaller ones, have a tinted plastic cover over the face of the picture tube. Usually, these covers are removable. You should remove the plastic screen cover to allow the light pen to come in close contact with your video screen.

With a larger screen TV or video monitor, it is possible to write programs for higher resolution graphics. We have kept our six programs on cassette limited to low resolution graphics so they can work on any size screen.

BRIGHTNESS AND CONTRAST - The brightness and contrast settings of your TV will affect the readings of the light pen. Best operation will be achieved with a good amount of contrast and average brightness. You can experiment with TV settings by using the 'TV Adjust' program included on the enclosed cassette.

Room lighting is not usually critical, although a very bright room may have some adverse effect on light pen operation.

PROGRAMS ON CASSETTE

Included with your Colorware Light Pen is a cassette tape of six programs. All six programs are contained within a single Master Menu program. Simply CLOAD the cassette and type RUN. Select the program you desire by holding the light pen up to one of the six flashing boxes.

The light pen programs on cassette require 16K of memory. Extended Basic is NOT needed. Some programs include sound. The six programs represent a good sample of some of the things that can be done with the light pen and some imaginative programming.

HANGMAN - A light pen version of the classic word game. You can change the selection of words by altering lines 1090 and 1100.

CONCENTRATION - A game of memory skill. Try to match pairs of letter boxes. Each match adds 10 points to your score. For every miss, you lose 5 points.

QUIZ - A multiple choice type quiz program. You can write your own quiz by changing the question on line 3280 and the answers on lines 3120-3210.

DRAW - Allows you to draw a low resolution screen picture, block by block. Simply position the tip of your light pen where you want a white block to appear. You may halt the drawing process at any time and review your drawing by pressing any key. To resume drawing, hit any key again.

BUG CHASE - Is a Game of speed and coordination. Try to catch the lightening bugs one at a time before they disappear. The more you catch, the higher your score.

TV ADJUST - Is a self explaining program that lets you check and set your TV brightness & contrast controls for optimum light-penning. TV adjustment is not critical. Any picture with a good amount of contrast should allow the light pen to function well.

HOW THE LIGHT PEN WORKS

The tip of the light pen contains a light sensitive photo cell. The photo cell is connected to the analog-to-digital converter in your computer via the Joystick 0 input. Any time the computer encounters the statement JOYSTK(0) it automatically takes a reading of the light pen's current value. Type-in and RUN this illustrative program to familiarize yourself with light pen readings:

```
10 CLS
20 PRINT@ 0, JOYSTK(0)
30 GOTO 20
```

If you cover the tip of the light pen with your finger you should get a zero or near zero reading. Aim the light pen at a light source and you should read somewhere between 5 and 63, depending on the brightness of the light.

The light pen reads the amount of light that passes through its front aperture and strikes the photo cell within. Now place the pen over a dark portion of your TV screen and compare that reading to the one you get when you place it over a bright area on the screen. The fact that the light pen can recognize the difference between a light spot and a dark spot is the basis of all light pen programming.

It is important to note that the light pen can not truly read the color from your screen, but rather, the amount of light. To a limited degree, you can sense color because different colors emit different amounts of light. Green, black and white for example, each contain a different amount of light and would each give a different light pen value. It would be difficult however, to get a reliable reading difference between colors such as blue, red and violet.

LIGHT PEN PROGRAMMING

The foundation of light pen programming is scanning. You could for example, starting with a black screen, turn each individual screen graphic block white and then black again in rapid succession - reading the light pen with each step. When the particular graphic block at the location of the light pen goes white, the light pen will suddenly read noticeably higher. By knowing which graphic block effected the light pen output, the computer can ascertain the light pen's screen position.

This basic theme can be varied. Suppose you wanted to use the light pen to select from a screen menu of four items. This example program below accomplishes that. We explain how the program works on the next page for your clarification.

```
10 CLS
20 FOR N=1 TO 4
30 PRINT@N*64, CHR$(207)+CHR$(207);
40 FORDD=1 TO 10: NEXTDD
50 J(N)=JOYSTK(0)
60 PRINT@ N*64, CHR$(240)+CHR$(240);
70 NEXT N
80 AV=( J(1)+J(2)+J(3)+J(4) ) / 4
90 Z=0
100 FOR N=1 TO 4
110 IF J(N) > AV+3 THEN AV=J(N): Z=N
120 NEXT N
130 IF Z=0 THEN 20
140 D=JOYSTK(0)
150 PRINT@ Z*64, CHR$(207)+CHR$(207);
160 FOR DD=1 TO 10: NEXT DD
170 IF JOYSTK(0) < D+5 THEN 20
```

+ 500K

LINES 20 & 70 set up the scanning loop for 4 elements.

LINE 30 'turns-on' a graphic box by printing two white character blocks. If your TV screen is larger than 9", you only have to use one character block.

LINE 40 is a short delay loop to give the light pen time to register the light from the white block.

LINE 50 reads the joystick port 0 (where the light pen is connected) and stores the reading as J(N).


LINE 80: After all 4 readings are taken and stored, the four values are averaged to give us a starting reference point.

LINES 100, 110, 120 form a loop which determines which of the four was the highest. In order to keep errors to a minimum, we set it so J(N) has to be at least 3 higher than the average (AV) of our readings. If a reading is found to be higher than AV+3 then it is considered to be the selected one unless on a subsequent pass through this loop, another J(N) reads even higher.

When one is found to read the highest so far, the variable Z is set to N, thus storing the number of the highest reading.

LINE 130: If Z still equals zero at this point it means that no reading was higher than AV+3 and scanning is resumed at line 20.

LINES 140-170 are a double check system. The pen is read first while the selected box is dark (D), then turned-on (white) and read again. If the white reading is at least 5 greater than the dark reading then we have found the right box, otherwise, we resume scanning at line 20. This double check system is necessary to prevent false triggering and we recommend that you use it in all of your programs.



COLORWARE INC.
78-03 Jamaica Ave.
Woodhaven, NY 11421

